

# Get your SAS in gear – Automate Appendix Numbers, Titles and Footnotes

Liz Taylor, Endo Pharmaceuticals, Chadds Ford, PA

## **ABSTRACT**

When deadlines are looming it's time consuming, as well as frustrating, to have to make last minute 'cosmetic' programming changes such as:

- Reordering the Appendices to accommodate for the insertion/deletion of a Table/Listing/Figure
- Modifying Titles and/or Footnotes to correct misspellings or to include additional explanations that could not have been anticipated until the data was unblinded

Defining Appendix Numbers, Titles, and Footnotes separate from SAS, such as in an Excel spreadsheet, allows the SAS programmer to concentrate on generating the body of the Table/Listing/Figure without having to worry about making modifications to the program every time a change to the Titles and/or Footnotes is required.

Responsibility for the maintenance of the Title and Footnote information can even be delegated to a non-programmer. Titles and Footnotes can be 'previewed' prior to running the SAS program.

## **INTRODUCTION**

This paper discusses how to automate Appendix Numbers, Titles and Footnotes by creating a Dataset containing Title and Footnote information and using that information to define Macro Variables from which Titles and Footnotes can be generated and used by a Table/Listing/Figure (T/L/F) program.

### **Important considerations:**

- How to Store Title and Footnote Information
  - What is the Context and Format for my Titles/Footnotes?
  - How and Where should Title/Footnote Information be stored?
  - What Information do I need to store?
- How to Copy Title and Footnote Information into SAS
  - How do I define a Macro Procedure that will Import the Title/Footnotes Information and Generate a permanent SAS File?
- How to Generate Titles and Footnotes
  - How do I define a Macro Procedure that will Automatically generate Title/Footnote Statements?
- How to Run the T/L/F Program
  - How do I integrate the Macro Procedures that will Automatically generate Titles/Footnotes into a T/L/F Program?

## How to Store Title and Footnote Information

### What is the Context and Format for my Titles/Footnotes?

Titles and Footnotes are usually a combination of Statements that are the same in every Study and Statements that are unique to the Study such as:

- Title 1 – Company Name
- Title 3 – Appendix Title
- Titles 4 thru n – Study Specific Titles
- Last Title – Underscores preceded by a blank line
- Footnote 1 – Underscores followed by a blank line
- Footnotes 3 thru n – Study Specific Footnotes
- Last Footnote – Program Name, Appendix Name and Date preceded by a blank line (if possible)

Endo Pharmaceuticals Protocol Id: EN1234-001							Page 1 of 12	T1
Appendix 16.4, Listing 3								T2
Patient Demographics and Screening Characteristics								T3
-----								T4
-----								T5
-----								T6
Patient ID	Gender	Age (Yrs)	Race	Date Informed Consent Signed	Date of Birth	Primary Source of Pain (N)=Neuropathic (C)=Cancer	Rating of Pain	Subject is...
-----								
100-001	MALE	64	BLACK	29AUG2003	20MAY1939	(N) DIABETIC NEUROPATHY (DN)	MODERATE	[1]
100-002	MALE	79	CAUCASIAN	03SEP2003	28JUN1924	(N) POST-HERPETIC NEURALGIA (PHN)	MODERATE	[2]
100-003	MALE	65	CAUCASIAN	12SEP2003	06APR1938	(N) IDIOPATHIC SENSORY NEUROPATHY	MODERATE	[1]
100-004	FEMALE	65	CAUCASIAN	16SEP2003	11AUG1938	(N) OTHER PERIPHERAL NEUROPATHY-BILAT PERONEAL NEUROPATHY	MODERATE	[1]
100-005	MALE	82	CAUCASIAN	17SEP2003	01JAN1921	(N) OTHER PERIPHERAL NEUROPATHY-POORLY CONTROLLED PERIPHERAL NEUROPATHY	MODERATE	[2]
-----								
[1] Suboptimally responding to Current Analgesic Regimen								F1
[2] Experiencing intolerable Side Effects								F2
								F3
								F4
								F5
								F6
Program: X:\EN1234\001\PGMS\L_DEMOG1.SAS Output: X:\EN1234\001\OUTPUT\L_DEMOG1A.DOC Final Date: 06DEC2005								

## How to Store Title and Footnote Information

### How and Where should Title/Footnote Information be stored?

Using Excel or MS-Access:

- A CSV File can easily be generated and copied into SAS
- Data entry is easy
- A non-SAS person can maintain the information

### What Information do I need to store?

A	B	C	D	E	F	G
app_id	app_name	pgm_name	tf_id	tf_algn	tf_len	tf_text
L03	l_demog1a	l_demog1	T01	L	44	Endo Pharmaceuticals Protocol ID: EN1234-001
L03	l_demog1a	l_demog1	T03		24	Appendix 16.4, Listing 3
L03	l_demog1a	l_demog1	T04		50	Patient Demographics and Characteristics
L03	l_demog1a	l_demog1	T06		132	_____
L03	l_demog1a	l_demog1	F01	L	132	_____
L03	l_demog1a	l_demog1	F03	L	57	[1] Suboptimally responding to Current Analgesic Regimen
L03	l_demog1a	l_demog1	F04	L	41	[2] Experiencing intolerable side effects
L03	l_demog1a	l_demog1	F06	L	112	Program: X:\EN1234\001\PGMS\l_demogl.sas Output: X:\EN1234\001\OPUTPUT\l_demog1a.doc

#### Col

- A app\_id – Appendix Identifier: Type (T=Table, L=Listing) & Number (2 characters)
- B pgm\_name – Program Name – same as source program
- C app\_name – Appendix Name – same as output document
- D tf\_id - Title/Footnote Id: Type (T=Title, F=Footnote) & Number (2 characters)
- E tf\_algn – Title/Footnote Alignment: (C or Blank=Centered, L=Left Justified)
- F tf\_len – Length of Text (automatically generated)
- G tf\_text – Title/Footnote Text

#### Notes:

- First row contains variable names that will be used by SAS – these must not be changed
- Use Mock-ups to determine Appendix Numbers, Program Names and Appendix Names
- Program Name and Appendix Name together must be unique for each T/L/F
- Every Title and Footnote is defined
  - It is not necessary to define blank lines
- The Last Footnote (containing Program and Output Names) needs to have Date added at time of execution
- The length of the Text is automatically generated making it easier to format Titles/Footnotes that require more than one line – other than that it is not used
  - Use the Length Function ( =len(Gx) )

## How to Copy Title and Footnote Information into SAS

**How do I define a Macro Procedure that will Import the Title/Footnotes Information and Generate a permanent SAS File?**

- 1) Read in Titles & Footnotes from Excel
  - Split Appendix ID to define Appendix Number and Appendix Type (Table/Listing/Figure)
  - Split Title & Footnote ID to define Title/Footnote Number and Type (Title or Footnote)
- 2) Create a Permanent SAS File

```
%macro tf_import;

  /* (1) Read in Titles & Footnotes from Excel */
  data app_mstr;

    infile 'E:\PharmaSUG 2006\app_mstr.csv' dlm=',',
          MISSOVER DSD lrecl=32767 firstobs=2;
    informat app_id $3. app_name $10. pgm_name $10. tf_id $3.
           tf_algn $1. tf_len best2. tf_text $132.;
    input app_id $ app_name $ pgm_name $ tf_id $
          tf_algn $ tf_len tf_text $;

    /* Split Appendix ID (app_id) to Define Appendix Number (app_num) & Type (app_type) */
    length app_num 8. app_type $1.;
    app_num = input(substr(app_id,2),2.); /* Appendix Type */
    app_type = substr(app_id,1,1); /* Appendix Number */

    /* Split Title & Footnote ID (tf_id) to Define Title & Footnote Number (tf_num) & Type
    (tf_type) */
    length tf_num 8. tf_type $1.;
    tf_num = input(substr(tf_id,2),2.); /* Title/Footnote Type */
    tf_type = substr(tf_id,1,1); /* Title/Footnote Number */
  run;

  /* (2) Create a permanent SAS File (app_mstr) */
  proc sort data=app_mstr out=adata.app_mstr;
    by app_id app_name pgm_name descending tf_type tf_num;
  run;

%mend;
```

## How to Generate Titles and Footnotes

**How do I define a Macro Procedure that will Automatically generate Title/Footnote Statements?**

- 1) Cancel any previous Titles & Footnotes
- 2) Define a Macro Variable for each possible Title & Footnote text and alignment and initialize each to empty
- 3) Read the Appendix Master and generate a Macro Variable for each Title & Footnote Text and Alignment associated with the T/L/F
  - Extract Titles & Footnotes that match Program & Appendix Name
  - Add Current Date to Last Footnote
  - Assign values to Macro Variables for Text & Alignment
- 4) Define each Title and Footnote Statement

## How to Generate Titles and Footnotes

```
%macro tf_add (pgm_name=, app_name=);

/* Parameters: */
/* pgm_name - Program Name as defined in Appendix Master */
/* app_name - Appendix Name as defined in Appendix Master */

/* (1) Cancel any previous Titles and Footnotes */
title;
footnote;

/* (2) Define Macro Variables title1-10, talgn1-10, footnote1-10, and falgn1-10 */
/* to contain Text & Alignment values for each Title & Footnote - Initialize each to empty */
%do i = 1 %to 10;
    %let title&i = ; /* Title Text */
    %let talgn&i = ; /* Title Alignment */
    %let footnote&i = ; /* Footnote Text */
    %let falgn&i = ; /* Footnote Alignment */
%end;

/* (3) Read the Appendix Master and generate a Macro Variable for each */
/* Title & Footnote Text and Alignment associated with the T/L/F */
data mstr;
    set adata.app_mstr;

    /* Extract Titles & Footnotes that match Program & Appendix Name */
    if upcase(trim(left(pgm_name)))=%upcase("&pgm_name") and
    upcase(trim(left(app_name)))=%upcase("&app_name");

    /* Add Current Date to Last Footnote */
    if tf_type = 'F' and tf_text =: 'Program:' then
        substr(tf_text,110) = "Final Date: " ||
        trim(left(put(date(),date9.)));

    /* Assign values to Macro Variables for Text & Alignment */
    if tf_type = 'T' then
        call symput('title' || trim(left(put(tf_num,2))), tf_text);
    if tf_type = 'F' then
        call symput('footnote' || trim(left(put(tf_num,2))), tf_text);
    call symput(trim(left(tf_type)) || 'algn' ||
        trim(left(put(tf_num,2))), tf_algn);
run;

/* (4) Define each Title & Footnote Statements */
/* Define Alignment - L = Left, C or Blank = Center */
%do i = 1 %to 10;
    /* Define Titles */
    %if &talgn&i = L %then
        %do;
            title&i "&&title&i"; /* Left Justify */
        %end;
    %if &talgn&i = C or &talgn&i = %then
        %do;
            title&i &&title&i; /* Right Justify */
        %end;

    /* Define Footnotes */
    %if &falgn&i = L %then
        %do;
            footnote&i "&&footnote&i"; /* Left Justify */
        %end;
    %if &falgn&i = C or &falgn&i = %then
        %do;
            footnote&i &&footnote&i; /* Right Justify */
        %end;
    %end;
%end;

%mend;
```

## How to run the T/L/F Program

How do I integrate the Macro Procedures that will Automatically generate Titles/Footnotes into a T/L/F Program?

```

/* Import Titles/Footnotes */
%tf_import;

/* Define Titles/Footnotes */
%tf_add (pgm_name=l_demog1, app_name=l_demog1a);

/* Generate Table/Listing */
proc report data=demog headline headskip split='#' spacing=1 nowindows missing;
  columns patid clinid genderc age race x infco_dx
         birth_dx source_x painca_d ques_x;
  define patid / order noprint;
  define clinid / computed width=7 'Patient ID';
  define genderc / width=6 'Gender';
  define age / width=5 ' Age ' '(Yrs)';
  define race x / width=9 flow 'Race';
  define infco_dx / width=9 'Date Informed Consent Signed';
  define birth_dx / width=9 'Date of Birth';
  define source_x / width=30 flow 'Primary Source of Pain'
        '(N)=Neuropathic (C)=Cancer';
  define painca_d / width=8 'Rating of Pain';
  define ques_x / width=7 flow 'Subject is...';
run;

```

Endo Pharmaceuticals Protocol Id: EN1234-001

Page 1 of 12

Appendix 16.4, Listing 3  
Patient Demographics and Screening Characteristics

Patient ID	Age Gender	Date Informed Consent Signed	Date of Birth	Primary Source of Pain (N)=Neuropathic (C)=Cancer	Rating of Pain	Subject is...
100-001	MALE 64	29AUG2003	20MAY1939	(N) DIABETIC NEUROPATHY (DN)	MODERATE	[1]
100-002	MALE 79	03SEP2003	28JUN1924	(N) POST-HERPETIC NEURALGIA (PHN)	MODERATE	[2]
100-003	MALE 65	12SEP2003	06APR1938	(N) IDIOPATHIC SENSORY NEUROPATHY	MODERATE	[1]
100-004	FEMALE 65	16SEP2003	11AUG1938	(N) OTHER PERIPHERAL NEUROPATHY-BILAT PERONEAL NEUROPATHY	MODERATE	[1]
100-005	MALE 82	17SEP2003	01JAN1921	(N) OTHER PERIPHERAL NEUROPATHY-POORLY CONTROLLED PERIPHERAL NEUROPATHY (PHN)	MODERATE	[2]

[1] Suboptimally responding to Current Analgesic Regimen  
[2] Experiencing intolerable Side Effects

Program: X:\EN1234\001\PGMS\L\_DEMOG1.SAS Output: X:\EN1234\001\OUTPUT\L\_DEMOG1A.DOC Final Date: 06DEC2005

## How can I further enhance this procedure to make it even more valuable?

### Automate Standard Titles and Footnotes

Automating the Standard Titles and Footnotes significantly reduces the number of Titles and Footnotes that have to be defined saving additional time and effort

Delete Standard Title & Footnote Statements or those that can be easily derived from the Spreadsheet leaving only Study Specific Titles/Footnotes, such as:

Company Name (T1), Appendix Number (T3), Underscores at bottom of Titles (T6),  
Underscores at top of Footnotes (F1), Program Name (F6)

leaving only Study Specific Titles and Footnotes

A	B	C	D	E	F	G
app_id	app_name	pgm_name	tf_id	tf_algn	tf_len	tf_text
L03	I_demog1a	I_demog1	T04		50	Patient Demographics and Characteristics
L03	I_demog1a	I_demog1	F03	L	57	[1] Suboptimally responding to Current Analgesic Regimen
L03	I_demog1a	I_demog1	F04	L	41	[2] Experiencing intolerable side effects

### Modify the %tf\_import Macro Procedure to generate Standard Titles & Footnote as well as Study Specific Titles & Footnotes

- 1) Read in Titles & Footnotes from Excel (Same as Step 1 in original Macro Procedure)
- 2) Define Macro Variables containing Compound & Protocol
- 3) Summarize Study Specific Title & Footnote Numbers to determine Last Title & Footnote Number
- 4) Generate a SAS Dataset containing Standard Titles & Footnotes
- 5) Combine Study Specific Titles & Footnotes with Standard Titles & Footnotes
- 6) Create Permanent SAS Dataset (Same as Step 2 in original Macro Procedure)

```
%macro tf_import;

/* (1) Read in Titles & Footnotes from Excel */
data app_stdy;

  infile 'E:\PharmaSUG 2006\app_mstr.csv' dlm=',',
         MISSOVER DSD lrecl=32767 firstobs=2;
  informat app_id $3. app_name $10. pgm_name $10. tf_id $3.
         tf_algn $1. tf_len best2. tf_text $132.;
  input app_id $ app_name $ pgm_name $ tf_id $
        tf_algn $ tf_len tf_text $;

  /* Split Appendix ID (app_id) to Define
     Appendix Number (app_num) & Type (app_type) */
  length app_num 8. app_type $1.;
  app_num = input(substr(app_id,2),2.); /* Appendix Type */
  app_type = substr(app_id,1,1); /* Appendix Number */

  /* Split Title & Footnote ID (tf_id) to Define
     Title & Footnote Number (tf_num)& Type (tf_type) */
  length tf_num 8. tf_type $1.;
  tf_num = input(substr(tf_id,2),2.); /* Title/Footnote Type */
  tf_type = substr(tf_id,1,1); /* Title/Footnote Number */
run;

/* (2) Define Macro Variables containing Compound (&cmpnd)
and Protocol (&prtcl)*/
%let cmpnd=EN1234;
```

```

%let prtcl=001;

/* (3) Summarize Study Specific Title & Footnote Numbers */
/* to determine Last Title and Footnote Number */
proc summary data=app_stdy nway;
  class app_type app_num app_id app_name pgm_name tf_type;
  var tf_num;
  output out=app_lst (drop=_type_ _freq_) max=tf_lst;
run;

/* (4) Generate a SAS Dataset containing Standard Titles & Footnotes */
data app_stnd;
  set app_lst;
  /* Generate Number, Text and Alignment for Standard Titles & Footnotes */
  length tf_algn $1. tf_text $132.;
  if tf_type eq 'T' then
  do;
    /* title1 (Company Name) */
    tf_num = 1;
    tf_text = "Endo Pharmaceuticals Protocol Id: &cmpnd.-&prtcl.";
    tf_algn = 'L';
    output;
    /* title3 (Appendix Number) */
    tf_num = 3;
    if tf_type = 'L' then
      tf_text = "Appendix 16.4, Listing " || trim(left(put(app_num,2)));
    if tf_type = 'T' then
      tf_text = "Appendix 16.2.2, Table " || trim(left(put(app_num,2)));
    tf_algn = 'C';
    output;
    /* last title (underscores) */
    tf_num = tf_lst + 2;
    tf_text = repeat('_', 132);
    tf_algn = 'L';
    output;
  end;
  if tf_type eq 'F' then
  do;
    /* first footnote (underscores) */
    tf_num = 1;
    tf_text = repeat('_', 132);
    tf_algn = 'L';
    output;
    /* last footnote (Program and Output Name) */
    tf_num = tf_lst + 2;
    tf_text = "Program:X:\&cmpnd.\NDA\&prtcl.\PGMS\" ||
              trim(left(pgm_name)) || ".SAS " ||
              "Output:X:\ :&cmpnd.\NDA\&prtcl.\OUTPUT\" ||
              trim(left(app_name)) || ".DOC";
    tf_algn = 'L';
    output;
  end;
run;

/* (5) Combine Study Specific Titles & Footnotes (app_stdy) and
Standard Titles & Footnotes (app_stnd) */
data app_mstr;
  set app_stdy
      app_stnd;
run;

/* (6) Create a Permanent SAS Dataset (app_mstr) */
proc sort data=app_mstr out=adata.app_mstr;
  by app_id app_name pgm_name descending tf_type tf_num;
run;

%mend;

```

## Preview Titles/Footnotes

Previewing Titles/Footnotes prior to generating the Table/Listing makes review and correction very easy saving additional time and effort.

Define a Macro Procedure that will automatically generate a Report with Titles/Footnotes

- 1) Create a Dummy SAS File containing only one record and one Variable
- 2) Execute %tf\_add with an appropriate Program and Appendix Name
- 3) Generate a Dummy Report

```
%macro tf_test (pgm_name=, app_name=);  
  
  /* (1) Create a Dummy SAS File containing only one record and one Variable */  
  data test_ds;  
    length test_var $80.;  
    test_var = "Program Name: &pgm_name / Appendix Name: &app_name";  
  run;  
  /* (2) Execute %tf_add with an appropriate Program and Appendix Name */  
  %tf_add (pgm_name=&pgm_name, app_name=&app_name);  
  
  /* (3) Generate a Dummy Report */  
  proc report data=test_ds headline headskip nowindows;  
    columns test_var;  
    define test_var /width=80 'Test Titles & Footnotes';  
  run;  
%mend;
```

Execute the Macro

```
/* Execute the Macro Procedure */  
%tf_test (pgm_name=l_demog1, app_name=l_demog1a);
```

```
Endo Pharmaceuticals Protocol Id: EN1234-001  
  
                          Appendix 16.2.2, Table 4  
                          Patient Demographics and Screening Characteristics  
  
-----  
Test Titles & Footnotes  
-----  
Program Name: l_demog1 / Appendix Name: l_demog1a  
  
-----  
[1] Suboptimally responding to Current Analgesic Regimen  
[2] Experiencing intolerable Side Effects  
  
Program: X:\EN1234\001\PGMS\L_DEMOG1.SAS Output: X:\EN1234\001\OUTPUT\L_DEMOG1A.DOC Final Date: 06DEC2005
```

## **CONCLUSION**

Automating Titles and Footnotes allows the SAS programmer to spend more time on processing the data and generating the analyses needed to produce Appendices

## **CONTACT INFORMATION**

Your comments and questions are valued and encouraged. Contact the author at:

Liz Taylor  
Endo Pharmaceuticals Inc.  
100 Endo Blvd.  
Chadds Ford, PA 19317

Phone: (610) 558-9800 x4268  
Email: [taylor.liz@endo.com](mailto:taylor.liz@endo.com)